

Linear FM Synthesis: Test Stimuli for Rapid Analysis of Auditory Neurodynamics

Stuart D. Washington and Jagmeet S. Kanwal*

Department of Physiology and Biophysics, Georgetown University Medical Center, 3970 Reservoir Road, NW, Washington DC, 20057, USA. *e-mail: kanwalj@georgetown.edu

Introduction

The first step in most experimental studies in sensory and cognitive neuroscience is the preparation of a set of test stimuli. For studying auditory processing at any level of the brain, two types of test stimuli are of key importance. These are categorized as basic acoustic patterns (BAPs) and species-specific communication sounds. BAPs include frequency modulations (or FMs) that are continuous changes in frequency as a function of time. FMs are key components of communication sounds of many species, including birds, whales, primates, bats and humans (Payne and McVay, 1971; Fenton, 1977; Marler and Pickert, 1984; Kanwal et al., 1994; Hauser, 1996; Tallal et al., 1993).

Auditory neurons typically exhibit tuning to one or more best frequencies (Kanwal et al., 1999). Inhibitory response areas generally skirt the excitatory response areas, especially in cortical neurons, and create sharply tuned neurons that exhibit relatively unpredictable responses to dynamic stimuli, such as FMs (Suga and Schlegel, 1973; Tian and Rauschecker, 1998, 2004). These responses may depend on stimulus duration, frequency range, and rate at which frequencies traverse their excitatory and inhibitory receptive fields. Because the inhibitory and excitatory response areas are not always symmetrical, FM direction is also an important determinant of the response of a neuron (Zhang et al. 2003). All of these parameters influence the timing and number of spikes that will be generated by a complex FM type stimulus. Therefore, for systematic studies of auditory neural dynamics, parameters such as duration and FM bandwidth (depth or frequency range) should be varied independently.

There are three classes of FM sounds: upward, downward, and bidirectional FMs. In each class, the FMs can be either linear or non-linear, which includes exponential, logarithmic and sinusoidal FMs. This report explains an algorithm and source code that will enable an experimenter to generate different types of linear FMs. The source code for this algorithm is written as command arguments in the SIGNAL v. 3.1 scripting language. SIGNAL (Engineering Design, Inc.) is a commercially available program designed for sound synthesis and analysis (Beeman, 1999).

Varying FM Bandwidth versus FM Slope

A complete description of a linear FM requires that at least three parameters be known: the duration and the

FM's start and end frequencies. The change in frequency over time ("t" in ms) defines the slope of a linear FM, such that $Slope = \frac{f(t_2) - f(t_1)}{(t_2 - t_1)}$, i.e., bandwidth/duration. In speech sounds, slope of the formant transition is equivalent to slope of an FM and may be measured in Hz/ms (Tallal et al., 1993).

To have a constant slope, the duration and depth must vary together. The depth and duration of an FM may be varied independently from the slope by nullifying oscillogram values that lie outside of the desired range. Depth may also be varied as a function of slope while the duration remains constant. The depth- and slope-varying algorithms described below vary the parameter being modified as a function of duration. The depth-varying algorithm keeps the slope constant and the slope-varying algorithm keeps the depth constant (Fig. 1A). A third algorithm varies the slope and depth, while keeping the duration constant (Fig. 1B). This is useful for studying FM tuning of duration-tuned neurons (Cassiday et al., 1994; He et al., 1997).




Figure 1. Results of varying (A) the depth and (B) the slopes of frequency modulations. Starting from the top, the 2nd, 5th, 8th, 11th, and 14th FMs generated by each program are presented. The axes were set to 70 ms to better illustrate changes in *depth* and *slope*. All FMs have a 10 ms onset latency.

Program output

Examples of output from the FM depth versus FM slope varying routines of the program are shown in figure 1. The program "GENFM" generates 14 linear FMs with each loop iteration changing either the depth or duration by a step size of 1/13th or 7.7%.

```

INIT
    ! Allocates memory; improves program flow

INPUT
    ! (Input) Takes user input from keyboard
    ! Also creates a null-sound control

DECRMNT
    ! Sets initial values for diminishing variables

doloop 100 uvar1 2 14
    ! Loop starts here; all parameter values hereafter will
    ! vary based on user input

ENDPNPTS
    ! Endpnpts, here, defines start and end frequencies if
    ! the slope varies by depth

BUILDFM
    ! Creates the oscillogram of an FM based on user-
    ! defined parameters

NULLIFY
    ! Reduces all values that are outside of the depth
    ! range to zero.

TAPR
    ! Tapers oscillogram based on user input

NORM
    ! Normalizes frequency and time axes (optional)

INCRMNT
    ! Increases the decrement step size for slope or depth
    ! per loop iteration

SPCTGRM
    ! Performs an FFT on the oscillogram and generates a
    ! spectrogram.
enddo 100
  
```

SUBROUTINES

```

INIT
set ulevel advan
    ! Bypasses user interface and allows
sw echo off
    ! program to operate without interruption
bmem
    ! Allocates memory to frequency and time buffers.
1 64
14 32
0 0
1 32
14 512
  
```

Algorithm and source code

The SIGNAL script presented in Figure 2 generates linear FMs of varying depth and slope. Since it is traditionally accepted that auditory neurons respond preferentially to at least a single best frequency, the program is designed to center its FMs around a user specified frequency.

```

INPUT
text 1
Enter Sample Rate:
acc uvar5
*

text 1
Enter Center Frequency:
acc uvar2
*

text 1
Enter Depth:
acc uvar3
*

ass uvar3 = uvar3 % 2
ass uvar73 = uvar3 % uvar4
ass ulab4 = "slope =" & str uvar73
typ ulab4

text 1
Enter Duration:
acc uvar4
*

text 1
Enter Taper Length (rise/fall time)
acc uvar 71
*

set taplen uvar71

text 1
Upward or Downward FM ('u' or 'd'):
acc ulab1
*

text 1
Vary Depth ('d') or slope ('s'):
acc ulab2
*

if ulab2 EQ "d" then
ass ulab3 = "Non"
endif

if ulab2 EQ "s" then
text 1
Vary by depth ('f') or duration ('t'):
acc ulab3
*
endif

fill t 1
uvar5
0
sinv t 1
xft t 1 1
gt ft 1

DECRMNT
ifi ulab2 EQ "d" then
ass uvar8 = uvar4 % 28
ass uvar9 = uvar4 - uvar8
ass uvar10 = uvar8
endif

if ulab3 EQ "t" then
ass uvar8 = uvar4 % 14
ass uvar72 = 0
endif

ENDPNPTS
ass uvar6 = uvar2 - uvar3
ass uvar7 = uvar2 + uvar3
if ulab1 EQ "d" then
ass uvar6 = uvar2 + uvar3
ass uvar7 = uvar2 - uvar3
endif

BUILDFM
linramp2 t uvar1
uvar5
uvar6
uvar7
uvar4
sinv t uvar1

NULLIFY
if ulab2 EQ "d" then
keep t uvar1
uvar8
uvar9
endif

TAPR
taper t uvar1 20 21 22

NORM
ins t uvar1
b
10
0
if ulab2 EQ "d" then
ass uvar72 = uvar8 * 2
endif
if ulab3 NE "f" then
ins t uvar1
e
uvar72
0
endif

if ulab3 EQ "t" then
ass uvar72 = uvar72 + uvar8
endif

INCRMNT
if ulab2 EQ "d" then
ass uvar8 = uvar8 + uvar10
ass uvar9 = uvar9 - uvar10
endif

if ulab3 EQ "t" then
ass uvar4 = uvar4 - uvar8
endif

if ulab3 EQ "f" then
ass uvar3 = uvar3 - uvar8
endif

SPCTGRM
xft t uvar1 uvar1
gt ft uvar1
  
```

Figure 2. Command line source code for "GENFM"

The center frequency chosen for the FMs in figures 1A and 1B is 58 kHz. The duration was set to 60 ms and the FM's direction was set to 'd' or downward. The depth was set to 80 kHz and the taper length was set to 3.5 ms. Since the FM is downward, the FM start frequency was 98 kHz and the end frequency was 18 kHz. The highest frequency sampled by the FM (here, 98 kHz) must be less than 40% of the sample rate; so the sample rate was set to 250 kHz. The functionality of the subroutines and major sections of code are summarized below.

INIT (Initialization): This subroutine allows the user to bypass the command menu, user screens, command file execution reporting, and extended error messages. It allocates memory for fourteen of both 32 Kpt time buffers and 512 Kpt frequency-time buffers that are necessary to run the program.

INPUT: -queries the user and accepts user input from the keyboard. The user is asked to set the center frequency, depth, sample rate, duration, taper length (rise/fall time) and direction of FMs. The default units for frequency and time are Hz and ms, respectively. Next, the user decides if the FMs will vary by depth or by slope. If the user decides to vary the slope, the user is asked whether the slope should vary by depth or duration. This routine also generates a blank sound file to be used as a control.

DECRMNT (Decrement): -sets the nullification, duration, or depth decrement values for depth-varying and slope-varying portion of the program, respectively. In this program, the loop is set to decrease each of these 3 values by 1/14th. While "Decrmnt" sets the initial values, "Incrmnt" updates these values (see below).

LOOP: -generate FMs. All values set hereafter are variables that will change throughout the loop's iterations. All values set prior to this are either static values (i.e., they do not change) or they are initial values of variables.

ENDPNTS (Endpoints): -sets the FM's depth based on user input for values of start and endpoints. It adds to and subtracts from the center frequency one half of the entered value. The two resulting values are saved as two different user-variables. These two variables are the start and end frequencies. Depending on the user-defined FM direction ('up' or 'down'), either the greater or lesser of the two user-variables may be the FM's starting frequency.

BUILDFM (Build the FMs): - generates FMs based on user-defined parameters. The SIGNAL command "linramp" (found in the directory "c:\signal\sigcom") simplifies linear FM synthesis and uses the parameters of sample rate, start frequency, end frequency, and

duration to create a waveform (or oscillogram) of an FM. However, "linramp" cannot be called from within another SIGNAL script. We therefore made a copy of "linramp" and saved a modified version, "linramp2". In "linramp2", "val L1" on lines 83 and 86 was replaced with "uvar5" (user-variable number 5), on lines 92 and 95 with "uvar6", on lines 104 and 107 with "uvar7", and on lines 116 and 119 with "uvar4".

NULLIFY: -nullifies oscillogram values that are outside of the desired depth range and works only when *depth* is varied and *slope* is held constant.

TAPR (Tapering): -contains a SIGNAL command to taper the oscillogram according to the settings described above.

NORM (Normalize): -scales the time-axis based on the duration of an oscillogram and delays sound onset by 10 ms.

INCRMNT (Increment): -increases the values set in "Decrmnt" by 1/13th per each loop iteration. The result is that the FMs decrease in either *duration* or *depth* per each loop iteration.

SPCTGRM (Spectrogram): -performs an FFT (fast-Fourier transformation) on the oscillogram and uses the results of that transform to generate a spectrogram. This is performed for each FM generated by this program.

A linear FM can also be generated from first principals by solving the following equation for a sinusoidal waveform, $S(t)$, of a carrier frequency f_{start} that is modulated over time " t ":

$$S(t) = A(t) \sin \left[2\pi f_{start} t + 2\pi \int_0^t m(t) dt \right]$$

where, $m(t) = slope \cdot t$

and, $slope = \frac{f_{end} - f_{start}}{t_{end} - t_{start}}$ for end and start frequencies (f) and times (t), respectively.

"A" is the waveform amplitude over each of three segments - increasing rapidly at the beginning (rise time), constant in the middle and decreasing at the end (fall time) of the FM sweep. This is equivalent to the TAPR routine in the SIGNAL code.

Acknowledgements

This project was supported in part by the National Institute on Deafness and other Communication Disorders (NIH Grant DC02054 to JSK).

References

Beeman, K., Digital Singal In: Hopp SL, Owren MJ, Evans CS (ed), Springer-Verlag, Heidelberg. pp. 59-103 (1998).

- Beeman, K. Belmont, Massachusetts: Engineering Design (1999).
- Fenton, M.B. Canadian Journal of Zoology 55,1151 (1977).
- Hauser, M.D. Cambridge, MA: MIT Press (1996).
- Kanwal, J.S., Fitzpatrick, D.C., Suga, N. *J Neurophysiol* 82, 2327 (1999)
- Kanwal, J.S., Matsumura, S., Ohlemiller, K.K., Suga, N. *J Acoust Soc Am* 96, 1229 (1994)
- Marler P, Pickert R (1984) *Animal Behavior* 32, 673-689.
- Ohlemiller, K.K., Kanwal, J.S., Suga, N. *Neuroreport* 7, 1749 (1996).
- Payne, R.S., McVay, S. *Science* 173, 585 (1971).
- Suga N, Schlegel PJ *Acoust Soc Am* 54, 174 (1973).
- He, J., Hashikawa, T., Ojima, H., Kinouchi, Y. *J Neurosci.* 17, 2615 (1997).
- Tian, B., Rauschecker, J.P. *J Neurophysiol.* 79, 2629 (1998).
- Tian, B., Rauschecker, J.P. *J Neurophysiol.* 92, 2993 (2004).
- Zhang, L.I., Tan, A.Y., Schreiner, C.E., Merzenich, M.M. *Nature* 424, 201 (2003).